

Computer Architecture Lab – CS322

Name: M. Maheeth Reddy

Roll No.: 1801CS31

Date: 29 August 2020

Lab 1 – Assembly Language Programming

Program 1:

Program to Swap 2 numbers

Specification: Swap 2 numbers that are stored at memory locations 2500 and 2501.

Source Code:

; Load value at memory address 2500 in register A

LDA 2500

; Copy value at A to register B

MOV B,A

; Load value at memory address 2501 in A

LDA 2501

; Store value at A in memory address 2500

STA 2500

; Copy value at A to B

MOV A,B

; Store value at A in memory address 2501

STA 2501

; Stop Execution

HLT

Input:

Memory Address	Value
2500	56
2501	44

Output:

Memory Address	Value
2500	44
2501	56

Program 2:

Program to count number of ones in the binary representation of a given number

Specification: The given number is assumed as DEC 95 and result is stored at memory location 2050.

Source Code:

```
; The given number is 95 (decimal)
MVI A,5F

; Copy value at register A to B
MOV B,A

; Register C contains remaining number of bits to be read
MVI C,08

; Register maintains the count of ones
MVI D,00

LOC:  RRC      ; Rotate bits to the right
      ; If Carry flag is 1 then increment D
      JNC LOC2
      INR D

LOC2:  DCR C    ; Decrement C
      JNZ LOC

MOV A,D
; Store the count at memory address 2050
STA 2050

; Stop Execution
HLT
```

Inputs and Outputs

Given Number	No. of 1's
95	06
01	01
FF	08

Program 3:

Program to count number of even numbers in given 10 numbers

Specification: Count number of even numbers among 10 numbers stored at memory addresses 2051 to 205A. Store the count at memory address 3050.

Source Code:

```
; Program to count number of even numbers in given 10 numbers

; Initialize H to memory address just before the address of first number
LXI H, 2050H

; Register C stores count
MVI C,00

; Register D stores remaining number of numbers to be checked
MVI D,0A

; For each number, check if it is even or odd
; If even, increment C and decrement D
; Otherwise just decrement D
LOOP: INX H
      MOV A,M
      ANI 01
      JNZ LOOP2
      INR C

LOOP2: DCR D
      JNZ LOOP

MOV A,C
; Store result at memory address 3050
STA 3050H
HLT
```

Input:

Memory Address	2051	2052	2053	2054	2055	2056	2057	2058	2059	205A
Value	13	14	15	16	17	16	15	14	13	01

Output:

Memory Address	3050
Value	04

Program 4:

Program to calculate sum of first N natural numbers

Specification: Calculate sum of first N natural numbers by the formula $\frac{N*(N+1)}{2}$. N is stored at memory address 2050 and result is stored at 2051.

Source Code:

; Value N is stored in register M which is pointed by register pair H,L
LXI H,2050

; Register A is used to calculate N*(N+1)
MVI A,00

; Register C is used to count number of times N has to be added to A
; It is initialized with (N+1) and decremented after every addition to A
MOV C,M
INR C

LOOP1: ADD M
DCR C
JNZ LOOP1

; Register C is used to calculate N*(N+1)/2 by repeatedly subtracting 2 from A
; The number of subtractions required to make A zero is the answer
MVI B,02H
MVI C,00H

LOOP2: INR C
SUB B
JNZ LOOP2

MOV A,C
; Store result at memory address 2051
STA 2051H
HLT

Input (N)	Output
04	0A
05	0F
08	24
0A	37

Program 5:

Program to find ASCII representation of a given BCD number

Specification: Find the ASCII representation of a BCD number stored at memory address 2000. Store the ASCII of Most Significant Byte (MSB) and Least Significant Byte (LSB) at memory addresses 3001 and 3000 respectively.

Source Code:

; Program to convert BCD to ASCII

LXI H, 2000H

MOV A,M

MOV B,M

; Do Bitwise AND of given number with 0F to get LSB

ANI 0FH

; BCD digits are only from 0-9

; ASCII values can be obtained by adding DEC 48 or HEX 30

ADI 30H

STA 3000H

; Do Bitwise AND of given number with F0

; and rotate left 4 times to get MSB

MOV A,B

ANI F0H

RLC

RLC

RLC

RLC

; Get ASCII of Most Significant Byte by adding DEC 48 or HEX 30.

ADI 30H

STA 3001H

HLT

Input (Value at 2000H)	Output (MSB at 3001H)	Output (LSB at 3000H)
23	32	33
51	35	31
99	39	39
07	30	37

Program 6:

Program to find BCD representation of an 8-bit number

Specification: Find the BCD representation of an 8-bit number stored at memory address 2050. Store the Most Significant Byte (MSB) and Least Significant Byte (LSB) at memory addresses 3051 and 3050 respectively.

Source Code:

; Program to convert HEX number to BCD

; Initialize the memory pointer

LXI H,2050H

; Clear Accumulator

MVI A,00

; Clear register D for MSB

MVI D,00

; Load HEX number to register C

MOV C,M

; Count the number one by one

LOOP2: ADI 01H

; Convert A to BCD format

DAA

JNC LOOP1

INR D

LOOP1: DCR C

JNZ LOOP2

; Store LSB at memory address 3050

STA 3050H

MOV A,D

; Store MSB at memory address 3051

STA 3051H

HLT

Input (Value at 2050H)	Output (MSB at 3051H)	Output (LSB at 3050H)
6F	11	01
51	00	81
FF	02	55
07	00	07

Program 7:

Program to find the largest number among 5 given numbers

Specification: Find the largest number among 5 numbers stored at memory addresses 2050-2054. Store the result at memory address 3050.

Source Code:

```
; Initialize Memory Pointer to Address of first number
LXI H, 2050H

; Register B stores number of numbers that are yet to be read
MVI B, 05H
DCR B

; Accumulator stores current maximum number
MOV A, M

; Read each number, compare it to current maximum
; If number is greater, update current maximum
LOOP:
    INX H
    CMP M
    JNC LOOP2
    MOV A, M
LOOP2:
    DCR B
    JNZ LOOP

; Store the final maximum number at address 3050
STA 3050H
HLT
```

Input:

Memory Address	2050	2051	2052	2053	2054
Value	15	20	06	12	04

Output:

Memory Address	3050
Value	20

Program 8:

Program to find Sum of N given numbers

Specification: A number N is stored at address 44FF. A list of N numbers is stored in memory starting at address 4500. Calculate the sum of these numbers and store it at address 4600. Store the final carry at address 4601.

Source Code:

```
; Sum of Series

; Value of N is stored at address 44FF
; It is loaded into register C
LDA 44FFH
MOV C,A

; Initialize Memory pointer with Address of first number
LXI H, 4500H

; Loop through each number
; Store Sum in A and Carry at B
MVI A, 00H
MVI B, 00H
LOOP:
    ADD M
    JNC LOOP2
    INR B
LOOP2:
    INX H
    DCR C
    JNZ LOOP

; Sum and Carry are stored at
; addresses 4600 and 4601 respectively
STA 4600H
MOV A,B
STA 4601H
HLT
```

Input

Memory Address	Value
44FF (N)	09
4500	20
4501	EA
4502	D8
4503	A3
4504	99
4505	80
4506	B7
4507	3C
4508	6D

Output

Memory Address	Value
4600 (Sum)	FE
4601 (Carry)	04

Program 9:

Program to find gray code of an 8-bit number

Specification: Find the gray code of an 8-bit number stored at memory address 2050. Store the result at memory address 3050.

Source Code:

```
; Load the value at address 2050 in Accumulator  
LDA 2050H
```

```
; Copy the value in A into B  
MOV B,A
```

```
; Reset Carry Flag  
STC  
CMC
```

```
; Rotate A by 1 bit to right  
RAR
```

```
; Perform XOR between original number  
; and rotated number to get gray code  
XRA B
```

```
; Store the result at address 3050  
STA 3050H  
HLT
```

Input	Output
45	67
FF	80
60	50
09	0D

Program 10:

Program to find 8-bit number from its gray code

Specification: The gray code of an 8-bit number is stored at address 2050. Find the number and store it at address 3050.

Source Code:

; Copy number to Accumulator
LDA 2050H

MVI C, 07H

; Copy value in A to B
MOV B,A

; Find Most Significant Bit of value in A
ANI 80H

; Perform the operations to calculate gray code
LOOP: RRC
 ANI 7FH
 XRA B
 DCR C
 JNZ LOOP

; Store Result at address 3050
STA 3050H
HLT

Input	Output
50	60
0D	09
67	45
80	FF